@echo off
rem ############################################################
rem # Nom      : import_file.bat                               #
rem # Creation : 2010-03-30                                    #
rem # Modif    : 2013-07-23                                    #
rem # Version  : 2.0                                           #
rem # Auteur   : DPI-CRR                                       #
rem # Env.     : Qualif                                        #
rem # ---------------------------------------------------------#
rem # Script d'import des fichiers                             #
rem ############################################################

setlocal EnableDelayedExpansion

REM ######################################
REM # Variables globales d'environnement #
REM ######################################

SET REP_RACINE=D:\gw2\gw2adm
D:
CALL %REP_RACINE%\profile.bat
CALL %SMID%\horodatage.bat


REM ###################
REM # Variable Locale #
REM ###################

SET appcode=gw2
SET returncode=0
SET logfile=%REP_TRACE%\import_file.log
SET seidimport=%SRVCFT%\import


REM #####################
REM # Boucle Principale #
REM #####################

FOR %%F IN (%seidimport%\*.%appcode%*) DO ( CALL :import "%%F" )
GOTO end

	
:import
REM Variable locale
SET myFic=%~1
SET checkdestfile=0
ECHO fichier = %myFic%

REM Boucle pour récupérer le filename et la taille
FOR %%i IN ("%myFic%") DO (
	SET filename=%%~nxi
	SET size=%%~zi
)
ECHO import : filename = %filename% \ size = %size%o

REM Vérifie qu un fichier destintation n est pas deja present
IF EXIST %REP_IMPORT%\%filename% ( SET checkdestfile=1 ) 

REM Deplacement du fichier
IF "%checkdestfile%"=="0" (
	MOVE /Y "%seidimport%\%filename%" "%REP_IMPORT%\%filename%"
	CALL :logfunction "[0] [taille !size!o] - Deplacement OK de %seidimport% vers %REP_IMPORT%"
) ELSE (
	REM Hordatage si fichier destination deja present
	MOVE /Y "%seidimport%\%filename%" "%REP_IMPORT%\%filename%.%ANNEE%%MOIS%%JOUR%%HEURE%%MIN%%SEC%"
	CALL :logfunction "[1] [taille !size!o] - Fichier deja present - Deplacement realise de %seidimport% vers %REP_IMPORT%\%filename%.%ANNEE%%MOIS%%JOUR%%HEURE%%MIN%%SEC%"
)
GOTO :EOF


REM #######
REM # Log #
REM #######

REM fonction pour ecrire dans la log
:logfunction
SET message=%~1
ECHO %ANNEE%-%MOIS%-%JOUR% %HEURE%:%MIN%:%SEC% : %filename% - %message% >> %logfile%
GOTO :EOF


REM ##########
REM # Sortie #
REM ##########
:end
EXIT %returncode%